
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc54362315]Corrections required
0.1 [bookmark: _Toc37249095][bookmark: OLE_LINK12][bookmark: OLE_LINK13][bookmark: OLE_LINK2][bookmark: OLE_LINK3][bookmark: OLE_LINK10][bookmark: OLE_LINK11][bookmark: OLE_LINK29][bookmark: OLE_LINK30][bookmark: _Toc54362316]function f_NR_RLC_Reconfiguration ()
	Function name
	f_NR_RLC_Reconfiguration ()

	Reason for change
	In the current TTCN implementation at Steps 8 and 16, the intention of the NR RRCReconfiguration message is to change the pollPDU and it is not required to include the other IEs such as MAC-CellGroupConfig, PhysicalCellGroupConfig and SPCellConfig, as their inclusion, especially, SPCellConfig would un-necessarily trigger a PRACH procedure from the UE which serves no purpose as the configuration of the SPCellConfig has remained the same as before. Attention is also drawn to the fact that in the SA version of this TC, the three IEs are not being included in the NR RRCReconfiguration message

	Summary of change
	Set the IEs MAC-CellGroupConfig, PhysicalCellGroupConfig and SPCellConfig to omit in the NR RRCReconfiguartion message.
NOTE: A Prose CR shall be raised at the next RAN5.

	TTCN module
	Common_NR_TC/7_1_2/RLC_TC_Common_NR.ttcn

	MCC160 Comment
	



Before Change:
	function f_NR_RLC_Reconfiguration(NR_CellId_Type p_CellId,
                                    integer p_NoOfNR_AMBearersN,
                                    integer p_NoOfNR_UMBearersM,
                                    template (value) NR_RLC_BearerToAddModList_Type p_RLC_BearerConfigList,
                                    template (omit) NR_RadioBearerList_Type p_SS_Drb_ConfigList := omit) runs on NR_BASE_PTC
  {
    if (f_GetTestcaseAttrib_ENDC(testcasename())) { //ENDC tc
      f_NR_ENDC_ReConfigAM_UM(p_CellId, p_NoOfNR_AMBearersN, p_NoOfNR_UMBearersM, p_SS_Drb_ConfigList, p_RLC_BearerConfigList, -, -, -, -, RRC_CONNECTED); // @sic R5s190800, R5s201010  sic@
    } else {
      f_NR5GC_ReConfigAM_UM(p_CellId, p_RLC_BearerConfigList);
    }
  }





[bookmark: OLE_LINK5][bookmark: OLE_LINK4]
After Change:
	    function f_NR_RLC_Reconfiguration(NR_CellId_Type p_CellId,
                                    integer p_NoOfNR_AMBearersN,
                                    integer p_NoOfNR_UMBearersM,
                                    template (value) NR_RLC_BearerToAddModList_Type p_RLC_BearerConfigList,
                                    template (omit) NR_RadioBearerList_Type p_SS_Drb_ConfigList := omit) runs on NR_BASE_PTC
  {
    if (f_GetTestcaseAttrib_ENDC(testcasename())) { //ENDC tc
      f_NR_ENDC_ReConfigAM_UM_7123x(p_CellId, p_NoOfNR_AMBearersN, p_NoOfNR_UMBearersM, p_SS_Drb_ConfigList, p_RLC_BearerConfigList, -, -, -, -, RRC_CONNECTED); // @sic R5s190800, R5s201010  sic@
    } else {
      f_NR5GC_ReConfigAM_UM(p_CellId, p_RLC_BearerConfigList);
    }
  }



0.2 [bookmark: _Toc54362317]New function f_NR_ENDC_ReConfigAM_UM_7123x
	Function name
	f_NR_ENDC_ReConfigAM_UM_7123x

	Reason for change
	In the current TTCN implementation at Steps 8 and 16, the intention of the NR RRCReconfiguration message is to change the pollPDU and it is not required to include the other IEs such as MAC-CellGroupConfig, PhysicalCellGroupConfig and SPCellConfig, as their inclusion, especially, SPCellConfig would un-necessarily trigger a PRACH procedure from the UE which serves no purpose as the configuration of the SPCellConfig has remained the same as before. Attention is also drawn to the fact that in the SA version of this TC, the three IEs are not being included in the NR RRCReconfiguration message

	Summary of change
	Set the IEs MAC-CellGroupConfig, PhysicalCellGroupConfig and SPCellConfig to omit in the NR RRCReconfiguartion message.
NOTE: A Prose CR shall be raised at the next RAN5.

	TTCN module
	Common/NR/NR_CommonFunctions.ttcn

	MCC160 Comment
	




	
function f_NR_ENDC_ReConfigAM_UM_7123x(NR_CellId_Type p_CellId,
                                   integer p_NoOfNR_AMBearersN,
                                   integer p_NoOfNR_UMBearersM,
                                   template (omit) NR_RadioBearerList_Type p_SS_Drb_ConfigList := omit,
                                   template (omit) NR_RLC_BearerToAddModList_Type p_RLC_BearerConfigList := omit,
                                   template (omit) SRB_ToAddModList p_SRB_ToAddModList := omit,
                                   template (omit) DRB_ToAddModList p_DRB_ToAddModList := omit,
                                   template (omit) SecurityConfig p_SecurityConfig := omit,
                                   template (omit) MAC_CellGroupConfig p_MAC_CellGroupConfig := omit,
                                   NR_RRC_STATE_Type p_RRC_State := RRC_IDLE,
                                   template (omit) PDSCH_ServingCellConfig.nrofHARQ_ProcessesForPDSCH  p_NrofHARQ_ProcessesForPDSCH := omit
                                   ) runs on NR_BASE_PTC // @sic R5s201010 sic@
  {
    // @sic R5-203533: added p_NrofHARQ_ProcessesForPDSCH sic@
    var DRB_Identity v_DrbId := tsc_NR_DRB2;    // for ENDC only
    var template (value) CellGroupConfig v_SCGConfig;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerConfigList;
    var template (value) DRB_ToAddModList v_DRB_ToAddModList;
    var template (value) SecurityConfig v_SecurityConfig;
    var template (value) NR_RadioBearerList_Type v_SS_Drb_ConfigList;
    var template (value) RLC_Config v_RLC_Config;
    var EUTRA_NR_Coordination_MSG v_Coordination_MSG_Value;
    var integer i;
    var boolean v_IsFR1:= f_NR_CellInfo_GetIsFR1(p_CellId);
    var template (omit) PDCP_Config.drb.statusReportRequired v_StatusReportRequired := true_;
    
    for (i:=0; i < p_NoOfNR_AMBearersN + p_NoOfNR_UMBearersM; i:=i+1) {
      if (i < p_NoOfNR_AMBearersN) {
        v_SS_Drb_ConfigList[i] := f_NR_GetRadioBearerConfig_FullAM(p_CellId, v_DrbId);
        v_RLC_Config := cs_38508_RLC_Config_AM_DRB(f_NR_UL_AM_RLC(v_IsFR1), f_NR_DL_AM_RLC(v_IsFR1));
      } else {
        v_SS_Drb_ConfigList[i] := f_NR_GetRadioBearerConfig_FullUM(p_CellId, v_DrbId); //ENDC function: Sdap=omit
        v_RLC_Config := cs_38508_RLC_Config_UM_DRB(f_NR_UL_UM_RLC(), f_NR_DL_UM_RLC(v_IsFR1));
        v_StatusReportRequired := omit;
      }
      v_RLC_BearerConfigList[i] := cs_38508_RLC_BearerConfig_DRB(v_DrbId, v_RLC_Config, cs_38508_LogicalChannelConfig_DRB);
      v_DRB_ToAddModList[i] := cs_38508_DRB_ToAddMod(v_DrbId, cs_38508_PDCP_Config (-, -, -, -, v_StatusReportRequired)); // @sic R5s200696, R5s200998 sic@
      v_DrbId := v_DrbId + 1;
    }
    
    v_SecurityConfig := cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig (f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), secondary);

    if (isvalue(p_SS_Drb_ConfigList)) {
      v_SS_Drb_ConfigList := valueof(p_SS_Drb_ConfigList);
    }
    if (isvalue(p_RLC_BearerConfigList)) {
      v_RLC_BearerConfigList := valueof(p_RLC_BearerConfigList);
    }
    if (isvalue(p_DRB_ToAddModList)) {
      v_DRB_ToAddModList := valueof(p_DRB_ToAddModList);
    }
    if (isvalue(p_SecurityConfig)) {
      v_SecurityConfig := valueof(p_SecurityConfig);
    }

    f_NR_SS_CommonRadioBearerConfig(p_CellId, v_SS_Drb_ConfigList);
    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(p_SRB_ToAddModList, v_DRB_ToAddModList, -, v_SecurityConfig);
    v_SCGConfig := f_NR_GetSCGConfigDef(p_CellId);

    v_SCGConfig.mac_CellGroupConfig :=  omit;
    v_SCGConfig.physicalCellGroupConfig := omit;
    v_SCGConfig.spCellConfig := omit;

    v_SCGConfig.rlc_BearerToAddModList := v_RLC_BearerConfigList;
    f_NR_SendRRCReconfigurationContentsToEUTRA(v_RadioBearerConfig, v_SCGConfig, -, -, false);  //@sic R5s190541, R5s200647 sic@

    // Get Security info from EUTRA, calculate latest keys and configure SS
    // as only need to do this for the first reconfiguration @sic R5s200647, R5s201010 sic@
    if (p_RRC_State != RRC_CONNECTED)
    {
      f_ENDC_GetKeysActivateSecurity(p_CellId, EUTRA);
    }

    //RRCReconfigurationComplete not yet received from E-UTRA - Wait for it @sic R5s200647 sic@
    EUTRA.receive(cmr_EUTRA_NR_OctetData ("Complete")) -> value v_Coordination_MSG_Value;
    f_NR_CheckReceived_RRCReconfigurationComplete(valueof(v_Coordination_MSG_Value));
  }







